Derivabilidad. Extremos de funciones. Gráficas

Los objetivos de esta práctica son:

• Analizar diversas órdenes de Maple que afectan a la derivación de funciones.
• Comprender el significado geométrico de la derivada.
• Hallar extremos relativos y absolutos de funciones.
• Representar gráficamente funciones reales de una variable real.

1. Derivación

1) Hallar la derivada de la función $f(x) = \frac{sen(x)+\cos(x)}{sen(x)-\cos(x)}$ y simplificar si es posible.

Calcular la derivada segunda.

En Maple, la orden para calcular la derivada de una función f en un punto x es **D(f)(x)** :

```
> restart:f:=x->(sin(x)+cos(x))/(sin(x)-cos(x)):
  D(f)(x);
```

$$\frac{\cos(x)-\sin(x)}{\sin(x)-\cos(x)} - \frac{(\sin(x)+\cos(x))^2}{(\sin(x)-\cos(x))^2}$$

Podemos calcularla en un punto concreto:

```
> D(f)(0);
```

$$-2$$

La orden **D(f)** devuelve la función derivada:

```
> D(f);
```

$$x \to \frac{\cos(x)-\sin(x)}{\sin(x)-\cos(x)} - \frac{(\sin(x)+\cos(x))^2}{(\sin(x)-\cos(x))^2}$$

Y también podemos presentarla más legible:

```
> 'D(f)(x)'=D(f)(x);
```

$$D(f)(x) = \frac{\cos(x)-\sin(x)}{\sin(x)-\cos(x)} - \frac{(\sin(x)+\cos(x))^2}{(\sin(x)-\cos(x))^2}$$

La derivada también se puede calcular con la orden **diff(f(x),x)** (obsérvese bien la sintaxis):

```
> diff(f(x),x);
```

$$\frac{\cos(x)-\sin(x)}{\sin(x)-\cos(x)} - \frac{(\sin(x)+\cos(x))^2}{(\sin(x)-\cos(x))^2}$$

En este segundo caso, no es necesario haber definido la función:

```
> diff((sin(x)+cos(x))/(sin(x)-cos(x)),x);
```

$$\frac{\cos(x)-\sin(x)}{\sin(x)-\cos(x)} - \frac{(\sin(x)+\cos(x))^2}{(\sin(x)-\cos(x))^2}$$

Hay algunas diferencias entre las dos órdenes. Las más importantes son:

a) Si queremos hallar la derivada de la función $g(x) = x^2$ no es correcto escribir **D(x^2)(x)**. Hay que definir antes la función $g(x) = x^2$ y luego escribir **D(g)(x)** .

b) Si queremos hallar la derivada de la función $g(x) = x^2$ en el punto $x = 3$, no es correcto escribir **diff(x^2,3)** , ni **diff(x^2,x)(3)** .

Intentemos simplificar la derivada. Podemos hacerlo con cualquiera de las dos órdenes:

```
> [simplify(diff(f(x),x)),simplify(D(f)(x))];
```

$$\left[\frac{2}{2\cos(x)\sin(x)-1}, \frac{2}{2\cos(x)\sin(x)-1}\right]$$

Podemos hallar derivadas de orden superior (derivada segunda, tercera, ...). La segunda se puede calcular de cualquiera estas formas: **(D@@2)(f)(x);** o **D(D(f))(x)** o **diff(f(x),x,x);** o **diff(f(x),x$2);** . Por ejemplo:

```
> diff(f(x),x$2);
  simplify((D@@2)(f)(x));
```

$$\frac{-\sin(x)-\cos(x)}{\sin(x)-\cos(x)} - \frac{3(\cos(x)-\sin(x))(\sin(x)+\cos(x))}{(\sin(x)-\cos(x))^2} + \frac{2(\sin(x)+\cos(x))^3}{(\sin(x)-\cos(x))^3}$$

$$\frac{4(\sin(x)+\cos(x))}{2\sin(x)\cos(x)^2+\sin(x)+2\cos(x)^3-3\cos(x)}$$

Para calcular la derivada segunda, puede ser conveniente derivar una vez, simplificar, volver a derivar la expresión simplificada y por último volver a simplificar. Hecho todo de una sola vez:

```
> simplify(D(simplify(D(f)))(x));
```

$$\frac{4(\sin(x)+\cos(x))}{2\sin(x)\cos(x)^2+\sin(x)+2\cos(x)^3-3\cos(x)}$$

Como vemos, en este ejemplo el resultado es el mismo que si no simplificamos después de hallar la primera derivada. Pero en otros casos puede dar resultados distintos (aparentemente, claro).

2. Significado geométrico de la derivada

2) Gráfica de la función $f(x) = x(x^2-1)$, de su tangente en $x = 0$ y de las cuerdas que pasan por $x = 0$ y por otro punto $d \to 0$-.

Esta es la explicación de las órdenes que hay más abajo:

• Borramos todas las asignaciones anteriores, para empezar desde el principio (**restart**).
• Cargamos el paquete "plots".
• Definimos la función f y el punto $a = 0$.
• Llamamos "grafdef" a la gráfica de f . La hacemos en el rango x=-1..1, y=-2/5..2/5, de color rojo y le ponemos un título. Pero de momento no dibujamos la gráfica (por eso, la orden termina en dos puntos, no en punto y coma).
• Llamamos "tangente" a la gráfica de la tangente a f en $x = a$, cuya ecuación es $f(a) + D(f)(a)(x-a)$. Aquí, $D(f)$ es la derivada de f. La hacemos de color verde y en el mismo rango x=-1..1 que la función. Tampoco la dibujamos, de momento.
• Ahora nos ocupamos de las cuerdas. La ecuación de la recta que pasa por el punto $(a, f(a))$ y por el punto $(d, f(d))$ es $y = f(a) + \frac{(f(d)-f(a))(x-a)}{d-a}$. Tomamos valores de d entre -1 y 0, es decir, son puntos a la izquierda de a. Tomamos 80 valores de d (**frames=80**) y las hacemos de color azul. De momento, no las dibujamos (la orden acaba en dos puntos). La orden es **animate** en lugar de **plot**, lo que hará que las 80 gráficas vayan apareciendo una tras otra para hacer un efecto de animación.

• Una vez que lo tenemos todo preparado, la orden **display** presenta todas las gráficas juntas. De momento, solo se ven la función y una recta. Piquemos en el dibujo; el menú de Maple cambia y aparecen diversas flechas (parada, avance, avance imagen a imagen, marcha adelante, marcha atrás, velocidad, ...). Pulsemos la flecha de avance.

```
> restart:with(plots):
  f:=x->x*(x^2-1):a:=0:
  grafdef:=plot(f(x),x=-1..1,y=-2/5..2/5,color=red,
  title="derivada en x=0 por la izquierda de f(x)=x(x^2-1)"):
  tangente:=plot(f(a)+D(f)(a)*(x-a),x=-1..1,color=black):
  cuerdas:=animate(f(a)+(f(d)-f(a))/(d-a)*(x-a),x=-1..1,
   d=-1..0,frames=80,color=blue):
  display([grafdef,tangente,cuerdas]);
```



3) Gráfica de la función $f(x) = 15\,x\,(x-1)\left(x-\frac{1}{2}\right)\left(x+\frac{2}{3}\right)(x+1)$, de su tangente en $x = 0$ y de las cuerdas que pasan por $x = 0$ y por otro punto $d \to 0+$.

Las órdenes son las mismas, con la única diferencia de que ahora queremos que las cuerdas se tomen con puntos d a la derecha de 0. Lo hacemos, por ejemplo, con valores de d desde $d = 1$ hasta $d = 0$. Si escribimos **d=1..0** nos da un error. Podemos escribir **d=0..1** y luego mover la animación marcha atrás. O también (y así está hecho) podemos tomar los puntos de la forma $1-d$, con valores de d desde $d = 0$ hasta $d = 1$.

```
> restart:with(plots):
  f:=x->15*x*(x-1)*(x-1/2)*(x+2/3)*(x+1):a:=0:
  grafdef:=plot(f(x),x=-1..1,y=-2..1,color=red,
  title="derivada en x=0 por la derecha de f(x)=15x(x-1)(x-1/2)
  (x+2/3)(x+1)"):
  tangente:=plot(f(a)+D(f)(a)*(x-a),x=-1..1,color=black):
  cuerdas:=animate(f(a)+(f(1-d)-f(a))/(1-d-a)*(x-a),x=-1..1,
   d=0..1,frames=80,color=blue):
  display([grafdef,tangente,cuerdas]);
```



4) Gráfica de la función $f(x) = \frac{1}{2} + 15x^2(x-1)\left(x-\frac{2}{5}\right)\left(x+\frac{2}{3}\right)(x+1)$, de su tangente en $x = 0$ y de las cuerdas que pasan por $x = 0$ y por otro punto $d \to 0$.

La única variación es que representamos simultáneamente las cuerdas a la izquierda y a la derecha.

```
> restart:with(plots):
  f:=x->1/2+15*x^2*(x-1)*(x-2/5)*(x+2/3)*(x+1):a:=0:
  grafdef:=plot(f(x),x=-1..1,y=-8/5..1,color=red,
  title="derivada en x=0 de f(x)=1/2+15x^2(x-1)(x-2/5)(x+2/3)
  (x+1)"):
  tangente:=plot(f(a)+D(f)(a)*(x-a),x=-1..1,color=black):
  cuerdasizda:=animate(f(a)+(f(d)-f(a))/(d-a)*(x-a),x=-1..1,
   d=-1..0,frames=80,color=blue):
  cuerdasdcha:=animate(f(a)+(f(1-d)-f(a))/(1-d-a)*(x-a),
   x=-1..1,d=0..1,frames=80,color=magenta):
  display([grafdef,tangente,cuerdasizda,cuerdasdcha]);
```



3. Extremos de funciones

5) Estudiar los extremos absolutos y relativos de $f(x) = \begin{cases} x^3 - x^2 - 8x + 1 & -2 \le x \\ 5\,e^{x+2} & \textit{en el resto} \end{cases}$

Definimos f y dibujamos la curva en algún intervalo:

```
> restart:
  f:=x->piecewise(-2<=x,x^3-x^2-8*x+1,5*exp(x+2));
```

$$f := x \to \mathit{piecewise}\left(-2 \le x, x^3 - x^2 - 8x + 1, 5\,e^{x+2}\right)$$

```
> plot(f(x),x=-4..4);
```



Como la gráfica sugiere, la función es continua también en –2:

```
> [limit(f(x),x=-2,left),limit(f(x),x=-2,right)];
```

$$[5, 5]$$

Calculamos los máximos y mínimos relativos estudiando la derivada de la función:.

```
> D(f)(x);
```

$$\begin{cases} 5\,e^{x+2} & x < -2 \\ \textit{undefined} & x = -2 \\ 3x^2 - 2x - 8 & -2 < x \end{cases}$$

Observamos que en el intervalo (x< –2) la derivada es siempre positiva, luego la función es creciente en $(-\infty,-2]$. Veamos lo que ocurre en el intervalo $[-2,\infty)$:

```
> solve({3*x^2-2*x-8>0});
```

$$\left\{x < -\frac{4}{3}\right\}, \{2 < x\}$$

Luego la función sigue siendo creciente en el intervalo [–2,–4/3], es decreciente en [–4/3,2] y es creciente en $[2,\infty)$. La función tiene un máximo relativo en x = –4/3 y un mínimo relativo en x = 2 . En el punto x = –2 , donde la función no es derivable, la función no tiene extremo relativo.

La función no tiene máximo absoluto, ya que

```
> limit(f(x),x=infinity);
```

$$\infty$$

La gráfica parece sugerir que hay un mínimo absoluto en x = 2. Y así es, ya que además del crecimiento y decrecimiento observado la función es positiva en $(-\infty,-2)$. El valor mínimo es:

```
> f(2);
```

$$-11$$

4. Representación gráfica de funciones

6) Estudiar y representar la función $f(x) = \dfrac{x^2 + 2}{x - 3}$ (dominio de definición, asíntotas, crecimiento y decrecimiento, extremos).

Para empezar, borramos todas las asignaciones que pudiera haber y definimos la función:

```
> restart:f:=x->(x^2+2)/(x-3);
```

$$f := x \to \frac{x^2 + 2}{x - 3}$$

Representemos la función. Su dominio es $R \setminus \{3\}$. En la gráfica siguiente hay que tener en cuenta que como se representa todo el eje horizontal (de $-\infty$ a ∞), la escala puede quedar distorsionada. Es decir, el aspecto será correcto a grandes rasgos, pero puede que los detalles no sean muy fieles.

```
> plot(f(x),x=-infinity..infinity);
```



La vertical corresponde a la asíntota $x = 3$. A continuación, podemos ver la gráfica en un rango más pequeño de x y de y (con la opción **discont=true** indicamos a Maple que la función puede no ser continua [esta opción no se puede usar si el rango de x es **x=-infinity..infinity**):

```
> plot(f(x),x=-15..25,y=-10..30,discont=true);
```



Averigüemos si la función tiene asíntotas horizontales u oblicuas:

```
> limit(f(x)/x,x=infinity);limit(f(x)-x,x=infinity);
```

$$1$$

$$3$$

De modo que $y = x + 3$ es asíntota de $f(x)$ cuando $x \to \infty$. Ahora veamos cuando $x \to -\infty$:

```
> limit(f(x)/x,x=-infinity);limit(f(x)-x,x=-infinity);
```

$$1$$

$$3$$

Así que la misma recta $y = x + 3$ es asíntota cuando $x \to -\infty$. Podemos representar la función y la asíntota juntas, la primera de color rojo y la segunda azul:

```
> plot([f(x),x+3],x=-15..25,y=-10..30,discont=true,
  color=[red,blue]);
```



Se advierte que la función tiene un máximo relativo cerca de 0 (¿quizá en 0?) y un mínimo relativo cerca de 6. Vamos a calcularlos mediante el estudio de la derivada de f (la función f es derivable):

```
> D(f)(x);
```

$$\frac{2x}{x-3} - \frac{x^2+2}{(x-3)^2}$$

Ahora simplifiquemos algo esa expresión, con las órdenes **simplify** o **factor**:

```
> factor(D(f)(x));
```

$$\frac{x^2-6x-2}{(x-3)^2}$$

El signo de la derivada:

```
> solve({x^2-6*x-2>0});
```

$$\{x < 3 - \sqrt{11}\}, \{3 + \sqrt{11} < x\}$$

Luego la función es estrictamente creciente en el intervalo cerrado $(-\infty, 3 - \sqrt{11}]$ y en el intervalo cerrado $[3 + \sqrt{11}, \infty)$. Y es estrictamente decreciente en el intervalo $[3 - \sqrt{11}, 3)$ y en el intervalo $(3, 3 + \sqrt{11}]$. Los puntos $3 - \sqrt{11}$ y $3 + \sqrt{11}$ son los extremos relativos que habíamos visto en la gráfica. Y no hay más extremos.

7) Estudiar en qué intervalos es cóncava y en cuáles es convexa la función $f(x) = \frac{e^x}{x}$.

Representemos la función. Su dominio es $R \setminus \{0\}$. Como siempre que se representa un intervalo no acotado de x, la escala puede estar distorsionada.

```
> restart:f:=x->exp(x)/x:
  plot(f(x),x=-infinity..infinity,thickness=2);
```



Estudiamos su concavidad y convexidad viendo el signo de la derivada segunda:

```
> factor(D(D(f))(x));
  solve({D(D(f))(x)>0});solve({D(D(f))(x)<0});
```

$$\frac{e^x (x^2 - 2x + 2)}{x^3}$$

$$\{0 < x\}$$

$$\{x < 0\}$$

La función es convexa en el intervalo abierto $(0, \infty)$. y cóncava en el intervalo abierto $(-\infty, 0)$. Observemos que en la gráfica anterior, no parece que sea convexa en $(0, \infty)$; es por la distorsión en la escala. Podemos hacer una gráfica algo más fiel quedándonos con un rango menor de x. Por ejemplo (la opción **scaling=constrained** fuerza a que la escala sea la misma en los dos ejes y la opción **discont=true** avisa a Maple de que la función pudiera no ser continua):

```
> plot(f(x),x=-5..8,y=-5..8,scaling=constrained,discont=true);
```



5. Más sobre gráficas de funciones

La orden `display`

Consideremos la función $f(x) = \frac{x^3}{(x+1)^2}$.

```
> restart:with(plots):f:=x->x^3/((x+1)^2):
```

El dominio es $R \setminus \{-1\}$. Su asíntota en $\pm\infty$ es $y = x - 2$. La función es cóncava en $(-\infty, -1)$ y en $(-1, 0)$, mientras que es convexa en $(0, \infty)$. Dibujemos la gráfica de la función y su asíntota:

```
> plot([f(x),x-2],x=-10..10,y=-10..10,scaling=constrained,
  discont=true);
```



Veamos otra forma de representar juntas varias funciones. Con estas órdenes Maple "calcula" las gráficas de la función y la asíntota y les da un nombre, pero no las presenta en la pantalla:

```
> funcion:=plot(f(x),x=-10..10,y=-10..10,scaling=constrained,
  color=red,discont=true):
> asint:=plot(x-2,x=-10..10,y=-10..10,scaling=constrained,
  color=blue):
```

Si ahora queremos verlas, podemos usar la orden `display`:

```
> display(funcion,asint);
```



Ahora, si queremos, podemos cambiar alguna cosa en la gráfica de la asíntota, sin rehacer la gráfica de la función. En el ejemplo siguiente, cambiamos el rango de x de la asíntota y su color. A continuación, presentamos la gráfica de f como antes y la de la asíntota con los cambios:

```
> asint:=plot(x-2,x=0..10,y=-10..10,scaling=constrained,
  color=green):
> display(funcion,asint);
```



Cada función ha salido representada en un rango distinto. En el siguiente ejemplo, representamos la función (en trazo más gordo) con dos colores: azul donde es convexa, y rojo donde es cóncava. También representamos la asíntota de color verde.

```
> concava1:=plot(f(x),x=-10..-1,y=-10..5,scaling=constrained,
  color=red,thickness=2):
> concava2:=plot(f(x),x=-1..0,y=-10..5,scaling=constrained,
  color=red,thickness=2):
> convexa:=plot(f(x),x=0..5,y=-10..5,scaling=constrained,
  color=blue,thickness=2):
> recta:=plot(x-2,x=-10..5,y=-10..5,scaling=constrained,
  color=green):
> display(concava1,concava2,convexa,recta);
```